Third Order Upwind scheme

> Aleksandar Donev, Courant Institute


```
> restart:
```

Formula for extrapolation from cell centers to cell faces for third-order upwind scheme:

```
> w_jph:=(w,j)->1/6*(-w(j-1)+5*w(j)+2*w(j+1));
```

$$w\_jph := (w,j) \mapsto -\frac{w(j-1)}{6} + \frac{5\,w(j)}{6} + \frac{w(j+1)}{3} \quad (1)$$

Finite Difference Interpretation

Finite difference interpretation of what w(j) is is just function evaluation:

```
> w_FD:=(j)->u(j*h);
```

$$w\_FD := j \mapsto u(j\,h) \quad (2)$$

Confirm that if advection is constant the finite difference is third order, so for constant advection ths is a third order scheme

```
> w_x_FD:=(w_jph(w_FD,0)-w_jph(w_FD,-1))/h;
```

$$w\_x\_FD := \frac{-u(-h) + \frac{u(0)}{2} + \frac{u(h)}{3} + \frac{u(-2\,h)}{6}}{h} \quad (3)$$

```
> series(w_x_FD, h, 5); # Third order
```

$$\mathrm{D}(u)(0) + \frac{1}{12}\,\mathrm{D}^{(4)}(u)(0)\,h^3 + \mathrm{O}(h^4) \quad (4)$$

Now consider space-dependent advection and write down the rhs of the ODEs in the spatial discretization:

```
> aw_x_FD:=(a((1/2)*h)*w_jph(w_FD,0)-a((-1/2)*h)*w_jph(w_FD,-1))/h;
```

$$aw\_x\_FD := \frac{1}{h}\left(a\left(\frac{h}{2}\right)\left(-\frac{u(-h)}{6} + \frac{5\,u(0)}{6} + \frac{u(h)}{3}\right) - a\left(-\frac{h}{2}\right)\left(-\frac{u(-2\,h)}{6} + \frac{5\,u(-h)}{6} + \frac{u(0)}{3}\right)\right) \quad (5)$$

Performing a Taylor series now shows an O(h^2) term:

```
> series(aw_x_FD, h, 4); # Only second order
```

$$a(0)\,\mathrm{D}(u)(0) + \mathrm{D}(a)(0)\,u(0) + \left(\frac{\mathrm{D}(a)(0)\,\mathrm{D}^{(2)}(u)(0)}{12} + \frac{\mathrm{D}^{(2)}(a)(0)\,\mathrm{D}(u)(0)}{8} + \frac{\mathrm{D}^{(3)}(a)(0)\,u(0)}{24}\right)h^2 + \mathrm{O}(h^3) \quad (6)$$

Finite Volume Interpretation

Now the interpretation of w is that it is an integral:

```
> w_FV:=(j)->Int(u(x),x=(j-1/2)*h..(j+1/2)*h)/h;
```

$$w\_FV := j \mapsto \frac{\int_{\left(j-\frac{1}{2}\right)h}^{\left(j+\frac{1}{2}\right)h} u(x)\,\mathrm{d}x}{h} \qquad \textbf{(7)}$$

Now write down the rhs of the ODE for FV:

```
> aw_x_FV:=(a((1/2)*h)*w_jph(w_FV,0)-a((-1/2)*h)*w_jph(w_FV,-1))/h;
```

$$aw\_x\_FV := \frac{1}{h}\left(a\left(\frac{h}{2}\right)\left(-\frac{\int_{-\frac{3h}{2}}^{-\frac{h}{2}} u(x)\,\mathrm{d}x}{6h} + \frac{5\left(\int_{-\frac{h}{2}}^{\frac{h}{2}} u(x)\,\mathrm{d}x\right)}{6h} + \frac{\int_{\frac{h}{2}}^{\frac{3h}{2}} u(x)\,\mathrm{d}x}{3h}\right) - a\left(-\frac{h}{2}\right)\left(-\frac{\int_{-\frac{5h}{2}}^{-\frac{3h}{2}} u(x)\,\mathrm{d}x}{6h} + \frac{5\left(\int_{-\frac{3h}{2}}^{-\frac{h}{2}} u(x)\,\mathrm{d}x\right)}{6h} + \frac{\int_{-\frac{h}{2}}^{\frac{h}{2}} u(x)\,\mathrm{d}x}{3h}\right)\right) \qquad \textbf{(8)}$$

Perform a series expansion of dw[0]/dt:

```
> dw0_dt_numerics:=convert(series(aw_x_FV, h, 6), polynom);
```

$$dw0\_dt\_numerics := a(0)\,\mathrm{D}(u)(0) + \mathrm{D}(a)(0)\,u(0) + \left(\frac{a(0)\,\mathrm{D}^{(3)}(u)(0)}{24} + \frac{\mathrm{D}(a)(0)\,\mathrm{D}^{(2)}(u)(0)}{8} + \frac{\mathrm{D}^{(2)}(a)(0)\,\mathrm{D}(u)(0)}{8} + \frac{\mathrm{D}^{(3)}(a)(0)\,u(0)}{24}\right)h^2 + \left(\frac{a(0)\,\mathrm{D}^{(4)}(u)(0)}{12} + \frac{\mathrm{D}(a)(0)\,\mathrm{D}^{(3)}(u)(0)}{12}\right)h^3 \qquad \textbf{(9)}$$

Now we need to compare this to the correct answer, which is itself an integral:

```
> dw0_dt:=Int(diff(a(x)*u(x),x),x=-h/2..h/2)/h;
```

$$dw0\_dt := \frac{\int_{-\frac{h}{2}}^{\frac{h}{2}} \left(\left(\frac{\mathrm{d}}{\mathrm{d}x} a(x)\right) u(x) + a(x)\left(\frac{\mathrm{d}}{\mathrm{d}x} u(x)\right)\right)\mathrm{d}x}{h} \qquad \textbf{(10)}$$

```
> dw0_dt_theory:=convert(series(dw0_dt,h,5),polynom); # Series expansion
```

$$dw0\_dt\_theory := a(0)\,\mathrm{D}(u)(0) + \mathrm{D}(a)(0)\,u(0) + \left(\frac{a(0)\,\mathrm{D}^{(3)}(u)(0)}{24} + \frac{\mathrm{D}(a)(0)\,\mathrm{D}^{(2)}(u)(0)}{8} + \frac{\mathrm{D}^{(2)}(a)(0)\,\mathrm{D}(u)(0)}{8} + \frac{\mathrm{D}^{(3)}(a)(0)\,u(0)}{24}\right)h^2 \qquad \textbf{(11)}$$

Now compute the truncation error, and see that it is now O(h^3), so this is third-order accurate as a FV scheme!

```
> simplify(dw0_dt_numerics-dw0_dt_theory); # Third order accurate!
```

$$\frac{\left(a(0)\,\mathrm{D}^{(4)}(u)(0) + \mathrm{D}(a)(0)\,\mathrm{D}^{(3)}(u)(0)\right)h^3}{12} \qquad \textbf{(12)}$$